



Integrated Circuits Design by FPGA

م.م. أحمد مؤيد عبدالحسين
جامعة الفرات الأوسط التقنية / الكلية التقنية الهندسية / نجف

Lecture 11

Packages and Components

Arithmetic Logic Unit (ALU)

Objectives of this Lecture

- To review **Packages** and **Components**
- To implement **ALU** example using **Components**.

Contents of this Lecture

- Introduction (Review)
- PACKAGE (Review)
- COMPONENT (Review)
- ALU Example using COMPONENTS

Introduction




Figure 10.1
Fundamental units of VHDL code.

Introduction

- Packages



- Components

Lecture 10 & 11

- Functions



- Procedures

Lecture 12 & 13





Introduction

- These new units can be located in the main code itself (that is, on the left-hand side of figure 10.1)
- However, since their main purpose is to allow common pieces of code to be reused and shared, it is more usual to place them in a **LIBRARY**.
- This also leads to code partitioning, which is helpful when dealing with long codes.
- In summary, frequently used pieces of code can be written in the form of **COMPONENTS**, **FUNCTIONS**, or **PROCEDURES**, then placed in a **PACKAGE**, which is finally compiled into the destination **LIBRARY**.
- We have already seen that at least two **LIBRARIES** are generally needed in a design: **ieee**, and **work**.
- After studying lectures 10 to 13, we will be able to construct our own libraries, which can then be added to the list above.

Introduction

- For example, commonly used circuits, like flip-flops, multiplexers, adders, logic gates, etc., can be placed in a LIBRARY, so any project can make use of them without having to explicitly rewrite such codes.

```
1  LIBRARY ieee;
2  USE ieee.std_logic_1164.all;
3  -----
4  ENTITY and2 IS
5  PORT (a, b: IN std_logic_vector(3 downto 0);
6        x: OUT std_logic_vector(3 downto 0));
7  END and2;
8  -----
9  ARCHITECTURE and2 OF and2 IS
10 BEGIN
11   x <= a and b;
12 END and2;
13 -----
```



PACKAGE

We start by describing the structure of a **PACKAGE**. Besides **COMPONENTS**, **FUNCTIONS**, and **PROCEDURES**, it can also contain **TYPE** and **CONSTANT** definitions, among others. Its syntax is presented below.

```
PACKAGE package_name IS  
    (declarations)  
END package_name;  
  
[PACKAGE BODY package_name IS  
    (FUNCTION and PROCEDURE descriptions)  
END package_name;]
```

As can be seen, the syntax is composed of two parts: PACKAGE and PACKAGE BODY

PACKAGE

- The first part (PACKAGE) is mandatory and contains all declarations, while the second part (PACKAGE BODY) is necessary only when one or more subprograms (FUNCTION or PROCEDURE) are declared in the first upper part, in which case it must contain the descriptions (bodies) of the subprograms.
- PACKAGE and PACKAGE BODY must have the same name.
- The declarations list can contain the following: COMPONENT, FUNCTION, PROCEDURE, TYPE, CONSTANT, etc.

PACKAGE

Example 10.1: Simple Package

The example below shows a PACKAGE called my_package. It contains only TYPE and CONSTANT declarations, so a PACKAGE BODY is not necessary.

```
1  -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  PACKAGE my_package IS
6      TYPE state IS (st1, st2, st3, st4);
7      TYPE color IS (red, green, blue);
8      CONSTANT vec: STD_LOGIC_VECTOR(7 DOWNT0 0) := "11111111";
9  END my_package;
10 -----
```

PACKAGE

- The next example (example 10.2) contains, besides TYPE and CONSTANT **declarations**, a FUNCTION. Therefore, a PACKAGE BODY is now needed (details on how to write a FUNCTION will be seen in lecture 12). This function returns TRUE when a positive edge occurs on clk.

PACKAGE

```
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 PACKAGE my_package IS
6     TYPE state IS (st1, st2, st3, st4);
7     TYPE color IS (red, green, blue);
8     CONSTANT vec: STD_LOGIC_VECTOR(7 DOWNT0 0) := "11111111";
9     FUNCTION positive_edge(SIGNAL s: STD_LOGIC) RETURN BOOLEAN;
10 END my_package;
11 -----
12 PACKAGE BODY my_package IS
13     FUNCTION positive_edge(SIGNAL s: STD_LOGIC) RETURN BOOLEAN IS
14     BEGIN
15         RETURN (s'EVENT AND s='1');
16     END positive_edge;
17 END my_package;
18 -----
```

PACKAGE

Any of the PACKAGES above (example 10.1 or example 10.2) can now be compiled, becoming then part of our *work* LIBRARY (or any other). To make use of it in a VHDL code, we have to add a new USE clause to the main code (USE work.my_package.all), as shown below.

```
-----  
LIBRARY ieee;  
USE ieee.std_logic_1164.all;  
USE work.my_package.all;  
-----  
ENTITY...  
...  
ARCHITECTURE...  
...  
-----
```

COMPONENT

- A **COMPONENT** is simply a piece of conventional code (that is, **LIBRARY** declarations + **ENTITY** + **ARCHITECTURE**, as seen in previous lectures).
- However, by declaring such code as being a **COMPONENT**, it can then be used within another circuit, thus allowing the construction of hierarchical designs.
- A **COMPONENT** is also another way of partitioning a code and providing code sharing and code reuse.

COMPONENT

- To use (instantiate) a COMPONENT, it must first be declared. The corresponding syntaxes are shown below.

COMPONENT declaration:

```
COMPONENT component_name IS
  PORT (
    port_name : signal_mode signal_type;
    port_name : signal_mode signal_type;
    ...);
END COMPONENT;
```

COMPONENT instantiation:

```
label: component_name PORT MAP (port_list);
```


COMPONENT

----- COMPONENT declaration: -----

```
COMPONENT inverter IS
```

```
    PORT (a: IN STD_LOGIC; b: OUT STD_LOGIC);
```

```
END COMPONENT;
```

----- COMPONENT instantiation: -----

```
U1: inverter PORT MAP (x, y);
```

COMPONENT




Figure 10.2 a. : Declarations in the main code itself

COMPONENT





Figure 10.2 b. : Declarations in a PACKAGE

ALU Example using COMPONENTS



ALU Example using COMPONENTS

sel	Operation	Function	Unit
0000	$y \leq a$	Transfer a	Arithmetic
0001	$y \leq a+1$	Increment a	
0010	$y \leq a-1$	Decrement a	
0011	$y \leq b$	Transfer b	
0100	$y \leq b+1$	Increment b	
0101	$y \leq b-1$	Decrement b	
0110	$y \leq a+b$	Add a and b	
0111	$y \leq a+b+cin$	Add a and b with carry	
1000	$y \leq \text{NOT } a$	Complement a	Logic
1001	$y \leq \text{NOT } b$	Complement b	
1010	$y \leq a \text{ AND } b$	AND	
1011	$y \leq a \text{ OR } b$	OR	
1100	$y \leq a \text{ NAND } b$	NAND	
1101	$y \leq a \text{ NOR } b$	NOR	
1110	$y \leq a \text{ XOR } b$	XOR	
1111	$y \leq a \text{ XNOR } b$	XNOR	

تتفيذ مباشر في فايل واحد ALU


```
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 USE ieee.std_logic_unsigned.all;
5 -----
6 ENTITY ALU IS
7     PORT (a, b: IN STD_LOGIC_VECTOR (7 DOWNTO 0);
8           sel: IN STD_LOGIC_VECTOR (3 DOWNTO 0);
9           cin: IN STD_LOGIC;
10          y: OUT STD_LOGIC_VECTOR (7 DOWNTO 0));
11 END ALU;
12 -----
13 ARCHITECTURE dataflow OF ALU IS
14     SIGNAL arith, logic: STD_LOGIC_VECTOR (7 DOWNTO 0);
15 BEGIN
16     ----- Arithmetic unit: -----
17     WITH sel(2 DOWNTO 0) SELECT
18         arith <=  a WHEN "000",
19                 a+1 WHEN "001",
20                 a-1 WHEN "010",
21                 b WHEN "011",
22                 b+1 WHEN "100",
```

```
23                 b-1 WHEN "101",
24                 a+b WHEN "110",
25                 a+b+cin WHEN OTHERS;
26     ----- Logic unit: -----
27     WITH sel(2 DOWNTO 0) SELECT
28         logic <=  NOT a WHEN "000",
29                 NOT b WHEN "001",
30                 a AND b WHEN "010",
31                 a OR b WHEN "011",
32                 a NAND b WHEN "100",
33                 a NOR b WHEN "101",
34                 a XOR b WHEN "110",
35                 NOT (a XOR b) WHEN OTHERS;
36     ----- Mux: -----
37     WITH sel(3) SELECT
38         y <=  arith WHEN '0',
39             logic WHEN OTHERS;
40 END dataflow;
41 -----
```

ALU Example using COMPONENTS

```
1  ----- COMPONENT arith_unit: -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  USE ieee.std_logic_unsigned.all;
5  -----
6  ENTITY arith_unit IS
7      PORT ( a, b: IN STD_LOGIC_VECTOR (7 DOWNTO 0);
8            sel: IN STD_LOGIC_VECTOR (2 DOWNTO 0);
9            cin: IN STD_LOGIC;
10           x: OUT STD_LOGIC_VECTOR (7 DOWNTO 0));
11 END arith_unit;
12 -----
13 ARCHITECTURE arith_unit OF arith_unit IS
14     SIGNAL arith, logic: STD_LOGIC_VECTOR (7 DOWNTO 0);
15 BEGIN
16     WITH sel SELECT
17         x <=  a WHEN "000",
18             a+1 WHEN "001",
19             a-1 WHEN "010",
20             b WHEN "011",
21             b+1 WHEN "100",
22             b-1 WHEN "101",
23             a+b WHEN "110",
24             a+b+cin WHEN OTHERS;
25 END arith_unit;
26 -----
```

```
1  ----- COMPONENT logic_unit: -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY logic_unit IS
6      PORT ( a, b: IN STD_LOGIC_VECTOR (7 DOWNTO 0);
7            sel: IN STD_LOGIC_VECTOR (2 DOWNTO 0);
8            x: OUT STD_LOGIC_VECTOR (7 DOWNTO 0));
9  END logic_unit;
10 -----
11 ARCHITECTURE logic_unit OF logic_unit IS
12 BEGIN
13     WITH sel SELECT
14         x <=  NOT a WHEN "000",
15             NOT b WHEN "001",
16             a AND b WHEN "010",
17             a OR b WHEN "011",
18             a NAND b WHEN "100",
19             a NOR b WHEN "101",
20             a XOR b WHEN "110",
21             NOT (a XOR b) WHEN OTHERS;
22 END logic_unit;
23 -----
```



ALU Example using COMPONENTS

```
1  ----- COMPONENT mux: -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY mux IS
6      PORT ( a, b: IN STD_LOGIC_VECTOR (7 DOWNTO 0);
7            sel: IN STD_LOGIC;
8            x: OUT STD_LOGIC_VECTOR (7 DOWNTO 0));
9  END mux;
10 -----
11 ARCHITECTURE mux OF mux IS
12 BEGIN
13     WITH sel SELECT
14         x <=  a WHEN '0',
15              b WHEN OTHERS;
16 END mux;
17 -----
```



ALU Example using COMPONENTS

```
1  ----- Project ALU (main code): -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY alu IS
6      PORT ( a, b: IN STD_LOGIC_VECTOR(7 DOWNTO 0);
7            cin: IN STD_LOGIC;
8            sel: IN STD_LOGIC_VECTOR(3 DOWNTO 0);
9            y: OUT STD_LOGIC_VECTOR(7 DOWNTO 0));
10 END alu;
11 -----
12 ARCHITECTURE alu OF alu IS
13 -----
14     COMPONENT arith_unit IS
15     PORT ( a, b: IN STD_LOGIC_VECTOR(7 DOWNTO 0);
16           cin: IN STD_LOGIC;
17           sel: IN STD_LOGIC_VECTOR(2 DOWNTO 0);
18           x: OUT STD_LOGIC_VECTOR(7 DOWNTO 0));
19     END COMPONENT;
20     -----
```

```
21     COMPONENT logic_unit IS
22     PORT ( a, b: IN STD_LOGIC_VECTOR(7 DOWNTO 0);
23           sel: IN STD_LOGIC_VECTOR(2 DOWNTO 0);
24           x: OUT STD_LOGIC_VECTOR(7 DOWNTO 0));
25     END COMPONENT;
26     -----
27     COMPONENT mux IS
28     PORT ( a, b: IN STD_LOGIC_VECTOR(7 DOWNTO 0);
29           sel: IN STD_LOGIC;
30           x: OUT STD_LOGIC_VECTOR(7 DOWNTO 0));
31     END COMPONENT;
32     -----
33     SIGNAL x1, x2: STD_LOGIC_VECTOR(7 DOWNTO 0);
34     -----
35 BEGIN
36     U1: arith_unit PORT MAP (a, b, cin, sel(2 DOWNTO 0), x1);
37     U2: logic_unit PORT MAP (a, b, sel(2 DOWNTO 0), x2);
38     U3: mux PORT MAP (x1, x2, sel(3), y);
39 END alu;
```

ALU Example using COMPONENTS




Figure 10.8

Simulation results of example 10.6.

Assignments

- The assignments will be attached to your class room.

End of lecture 11
Any Questions ?